NAG Fortran Library Routine Document

F11MMF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F11MMF computes the reciprocal pivot growth factor of an LU factorization of a real sparse matrix in compressed column (Harwell–Boeing) format.

2 Specification

```
SUBROUTINE F11MMF (N, ICOLZP, A, IPRM, IL, LVAL, IU, UVAL, RPG, IFAIL)
INTEGER            N, ICOLZP(*), IPRM(7*N), IL(*), IU(*), IFAIL
double precision   A(*), LVAL(*), UVAL(*), RPG
```

3 Description

F11MMF computes the reciprocal pivot growth factor $\max_j\left(\|A_j\|_\infty / \|U_j\|_\infty\right)$ from the columns A_j and U_j of an LU factorization of the matrix A,$P_rAP_c = LU$ where P_r is a row permutation matrix, P_c is a column permutation matrix, L is unit lower triangular and U is upper triangular.

4 References

None.

5 Parameters

1: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

2: ICOLZP($*$) – INTEGER array *Input*

On entry: ICOLZP(i) contains the index in A of the start of a new column. See Section 2.1.3 in the F11 Chapter Introduction.

3: A($*$) – ***double precision*** array *Input*

Note: the dimension of the array A must be at least $\mathrm{ICOLZP}(\mathrm{N}+1) - 1$, the number of non-zeros of the sparse matrix A.

On entry: the array of non-zero values in the sparse matrix A.

4: IPRM($7 \times \mathrm{N}$) – INTEGER array *Input*

On entry: the column permutation which defines P_c, the row permutation which defines P_r, plus associated data structures as computed by F11MEF.

5: IL($*$) – INTEGER array *Input*

On entry: records the sparsity pattern of matrix L as computed by F11MEF.

6: LVAL($*$) – ***double precision*** array *Input*

On entry: records the non-zero values of matrix L and some non-zero values of matrix U as computed by F11MEF.

7: IU($*$) – INTEGER array *Input*

On entry: records the sparsity pattern of matrix U as computed by F11MEF.

8: UVAL($*$) – ***double precision*** array *Input*

On entry: records some non-zero values of matrix U as computed by F11MEF.

9: RPG – ***double precision*** *Output*

On exit: the reciprocal pivot growth factor $\max_j\left(\|A_j\|_\infty/\|U_j\|_\infty\right)$. If the reciprocal pivot growth factor is much less than 1, the stability of the LU factorization may be poor.

10: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL $= 0$ unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, $\mathrm{N} < 0$.

IFAIL $= 2$

Ill-defined column permutations in array IPRM. Internal checks have revealed that the IPRM array is corrupted.

IFAIL $= 301$

Unable to allocate required internal workspace.

7 Accuracy

Not applicable.

8 Further Comments

If the reciprocal pivot growth factor, RPG, is much less than 1, then the factorization of the matrix A could be poor. This means that using the factorization to obtain solutions to a linear system, forward error bounds and estimates of the condition number could be unreliable. Consider increasing the THRESH parameter in the call to F11MEF.

9 Example

To compute the reciprocal pivot growth for the factorization of the matrix A, where

$$A = \begin{pmatrix} 2.00 & 1.00 & 0 & 0 & 0 \\ 0 & 0 & 1.00 & -1.00 & 0 \\ 4.00 & 0 & 1.00 & 0 & 1.00 \\ 0 & 0 & 0 & 1.00 & 2.00 \\ 0 & -2.00 & 0 & 0 & 3.00 \end{pmatrix}.$$

In this case, it should be equal to 1.0.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F11MMF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          LA, NMAX
      PARAMETER        (LA=10000,NMAX=1000)
      DOUBLE PRECISION ONE
      PARAMETER        (ONE=1.D0)
*     .. Local Scalars ..
      DOUBLE PRECISION FLOP, RPG, THRESH
      INTEGER          I, IFAIL, N, NNZ, NNZL, NNZU, NZLMX, NZLUMX,
     +                 NZUMX
      CHARACTER        SPEC
*     .. Local Arrays ..
      DOUBLE PRECISION A(LA), LVAL(8*LA), UVAL(8*LA)
      INTEGER          ICOLZP(NMAX+1), IL(7*NMAX+8*LA+4), IPRM(7*NMAX),
     +                 IROWIX(LA), IU(2*NMAX+8*LA+1)
*     .. External Subroutines ..
      EXTERNAL         F11MDF, F11MEF, F11MMF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F11MMF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
*
*     Read order of matrix
*
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read the matrix A
*
         DO 20 I = 1, N + 1
            READ (NIN,*) ICOLZP(I)
   20    CONTINUE
         NNZ = ICOLZP(N+1) - 1
         DO 40 I = 1, NNZ
            READ (NIN,*) A(I), IROWIX(I)
   40    CONTINUE

*
*        Calculate COLAMD permutation
*
         SPEC = 'M'
         IFAIL = 0
         CALL F11MDF(SPEC,N,ICOLZP,IROWIX,IPRM,IFAIL)
*
*        Factorise
*
         THRESH = ONE
         IFAIL = 0
```

```
         NZLMX = 8*NNZ
         NZLUMX = 8*NNZ
         NZUMX = 8*NNZ
         CALL F11MEF(N,IROWIX,A,IPRM,THRESH,NZLMX,NZLUMX,NZUMX,IL,
     +               LVAL,IU,UVAL,NNZL,NNZU,FLOP,IFAIL)
*
*        Calculate reciprocal pivot growth
*
         IFAIL = 0
         CALL F11MMF(N,ICOLZP,A,IPRM,IL,LVAL,IU,UVAL,RPG,IFAIL)
*
*         Output result
*
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Reciprocal pivot growth'
         WRITE (NOUT,'(F7.3)') RPG
*
      END IF
      END
```

9.2 Program Data

```
F11MMF Example Program Data
  5  N
 1
 3
 5
 7
 9
 12   ICOLZP(I) I=1,..,N+1
  2.   1
  4.   3
  1.   1
 -2.   5
  1.   2
  1.   3
 -1.   2
  1.   4
  1.   3
  2.   4
  3.   5    A(I), IROWIX(I) I=1,NNZ
```

9.3 Program Results

```
 F11MMF Example Program Results

 Reciprocal pivot growth
  1.000
```